# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=dhall-lsp-server
pkgver=1.1.2
pkgrel=82
pkgdesc="Language Server Protocol (LSP) server for Dhall"
url="https://github.com/dhall-lang/dhall-haskell/tree/master/dhall-lsp-server#readme"
license=("custom:MIT")
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-aeson' 'haskell-aeson-pretty' 'haskell-data-default' 'dhall'
         'dhall-json' 'haskell-lsp1.4' 'haskell-hslogger' 'haskell-lens' 'haskell-lens-family-core'
         'haskell-megaparsec' 'haskell-network-uri' 'haskell-optparse-applicative'
         'haskell-prettyprinter' 'haskell-rope-utf16-splay' 'haskell-unordered-containers'
         'haskell-uri-encode')
makedepends=('ghc' 'uusi' 'haskell-quickcheck' 'haskell-doctest' 'haskell-hspec'
             'haskell-lsp-test0.14.0' 'haskell-lsp-types1.4' 'haskell-tasty' 'haskell-tasty-hspec')
source=("https://hackage.haskell.org/packages/archive/$pkgname/$pkgver/$pkgname-$pkgver.tar.gz")
sha512sums=('af0fc8146ff87002069833624737fb0eb0f6ddb9250e9ef7e78a4b7e0d49b4071f707c5743714fecdc07abbf23507ec3a32e63e5fcfb3d3eebd612ddad46a2f8')

prepare() {
  cd $pkgname-$pkgver
  uusi -u lens $pkgname.cabal
}

build() {
  cd $pkgname-$pkgver    

  runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
    --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
    --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm \
    --ghc-option=-optl-Wl\,-z\,relro\,-z\,now \
    --ghc-option='-pie'

  runhaskell Setup build $MAKEFLAGS
  runhaskell Setup register --gen-script
  runhaskell Setup unregister --gen-script
  sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
  sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
  cd $pkgname-$pkgver
  LD_LIBRARY_PATH="$PWD/dist/build" PATH="$PWD/dist/build/dhall-lsp-server:$PATH" runhaskell Setup test --show-details=direct
}

package() {
  cd $pkgname-$pkgver

  install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
  install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
  runhaskell Setup copy --destdir="$pkgdir"
  install -D -m644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
  rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
